
/*************************************************************************** 

Copyright (c) 2017 Gizmo Garden LLC. This source code is distributed under 

terms of the GNU General Public License, Version 3, which grants certain 

rights to copy, modify, and redistribute. The license can be found at 

<http://www.gnu.org/licenses/>. There is no express or implied warranty, 

including merchantability or fitness for a particular purpose. 

***************************************************************************/ 

 

// Concept and design by Michael Lee. Enhancements by Bill Silver. 

 

int buttonPin   = 0;      // Digital input pin to read the button 

int pumpPin     = 1;      // Analog output (PWM) pin for pump 

int dialPin     = 1;      // Analog input pin for the dial (potentiometer) 

int pumpPower   = 255;    // 255 is max, 83 is the minimum 

 

int hoursToWait = 24;     // hours between waterings 

 

// Time at which the plant was last watered, in milliseconds. 

unsigned long lastWateringTime = 0; 

 

// Turn the pump on by ramping up power in steps to avoid current surge. 

void pumpOn() 

{ 

  for (int i = 1; i <= 10; ++i) 

  { 

    analogWrite(pumpPin, i * pumpPower / 10); 

    delay (20); 

  } 

} 

 

// Turn the pump off. 

void pumpOff() 

{ 

  analogWrite(pumpPin, 0); 

} 

 

// Read the dial to get time that the pump is on for, in milliseconds. Since 

// analogRead gets a value between 0 and 1023, and it's multiplied by 10, the 

// watering time will be somewhere between 0 and 10 seconds. 

void dialDelay() 

{ 

  delay(analogRead(dialPin) * 10); 

} 

 

void setup() 

{ 

  pinMode(pumpPin, OUTPUT); 

  pumpOff(); 

  pinMode(buttonPin, INPUT_PULLUP); 

} 

 

bool buttonIsPressed() 

{ 

  return digitalRead(buttonPin) == LOW; 

}  

void loop() 

{ 

  // If it's time to water the plant, do it 

  if (millis() - lastWateringTime >= hoursToWait * 3600000) 

  { 

    // Save the current time so we'll know when to water next 

    lastWateringTime = millis(); 

 

    // Turn the pump on for the time selected by the dial 

    pumpOn(); 

    dialDelay(); 

    pumpOff(); 

 

    // We're done for now 

    return; 

  } 

 

  // If the button is pressed, turn the pump on. Then wait for 0.25 sec 

  // and see if the button is still pressed. If so, leave the pump on 

  // for as long as the button is pressed. If not, run the pump for the 

  // duration set by the dial. 

  if (buttonIsPressed()) 

  { 

    pumpOn(); 

    delay(250); 

    if (buttonIsPressed()) 

      while (buttonIsPressed()) 

        delay(50); 

    else 

      dialDelay(); 

 

    // Now we can turn the pump off and we're done for now 

    pumpOff(); 

    return; 

  } 

 

  // Nothing happened, wait a bit and check again 

  delay(50); 

} 


